

Découvrir les données de data.bnf.fr à travers SPARQL

1 Documentation de data.bnf.fr

1.1 Naviguer dans data.bnf

Allez sur <http://data.bnf.fr>; consultez une page d'œuvre, d'auteur, de thème. Repérez les permaliens (en bas de page) et les données brutes en RDF/XML.

1.2 Allez sur le service SPARQL

Ouvrez votre navigateur à <http://data.bnf.fr/sparql>. Testez les requêtes pré-chargées. Regardez en particulier celle pour les dates biographiques d'un auteur.

1.3 Lecture du modèle de données

Allez à la page de documentation à <http://data.bnf.fr/opendata>. Prenez quelques minutes pour analyser cette doc.

Prenez le temps de consulter le diagramme de documentation du modèle de données. Téléchargez la documentation complète du modèle de données en cliquant sur le diagramme. Notez en particulier l'articulation entre :

- L'arborescence Work-Expression-Manifestation ;
- La différence entre « l'œuvre elle-même » (URI qui se termine par « #about ») et la notice de l'œuvre dans data.bnf.fr qui est un skos:Concept relié à l'œuvre par « foaf:focus » ;
 - Pourquoi le choix de skos:Concept pour la notice ?
- L'auteur d'une œuvre ;
- La version numérique de l'édition d'une œuvre ;

2 Interrogation de data.bnf.fr

2.1 Recherche dans les personnes

2.1.1 Première recherche

Recherchez une personne qui s'appelle « Jean de La Fontaine ». Le nom d'une personne est exprimée avec la propriété « foaf:name ». N'utilisez pas la fonction « regex ». Attention aux majuscules/minuscules !

Vous pouvez consulter les données RDF de sa fiche pour vous aider.

2.1.2 Désambiguisation

Que constatez-vous ? comment désambiguïser le résultat pour trouver "le bon" ? cliquez sur les URIs pour naviguer vers les fiches et trouver un critère différenciant ; regardez les données en RDF des fiches pour trouver les URIs des propriétés à utiliser. Ajoutez ce/ces critères à la requête.

Jean de la Fontaine = <http://data.bnf.fr/ark:/12148/cb11910267w#about>

2.2 Trouver toutes les œuvres écrites par Jean de La Fontaine

2.2.1 Liste des oeuvres

Enrichissez la requête pour trouver la liste des œuvres écrites par Jean de La Fontaine.

2.2.2 Et leur titre ?

Comment ramener le titre de ces œuvres ?

2.3 Trouver toutes les éditions de ces œuvres

2.3.1 Liste des éditions

Enrichissez la requête pour trouver les éditions des œuvres écrites par Jean de La Fontaine.

2.3.2 Et leur titre ?

Comment ramener le titre de ces éditions ? commentez la différence entre titre de l'œuvre et titre de l'édition.

2.3.3 Et leur ISBN ?

Comment ramener l'ISBN de ces éditions ? attention, toutes les éditions de ces œuvres n'ont pas un ISBN !

2.4 Trouver les versions numériques de ces éditions

2.4.1 La version numérique

Enrichissez la requête pour conservez les versions numériques de ces éditions. Attention, toutes les éditions n'ont pas une version numérique !

Commentez le rapport entre l'ISBN et la version numérique.

3 Solutions

3.1 Recherche dans les personnes

3.1.1 Première recherche

Recherchez une personne qui s'appelle « Jean de la Fontaine ».

```
SELECT ?personne
WHERE {
  ?personne a foaf:Person .
  ?personne foaf:name "Jean de La Fontaine" .
}
```

3.1.2 Désambiguisation

Que constatez-vous ? comment désambiguiser le résultat pour trouver "le bon" ?

Par exemple en sélectionnant les dates de naissance et de décès :

```
SELECT ?personne ?birth ?death
WHERE {
  ?personne a foaf:Person .
  ?personne foaf:name "Jean de La Fontaine" .
  ?personne bio:birth ?birth .
  ?personne bio:death ?death .
}
```

3.2 Trouver toutes les œuvres écrites par Jean de La Fontaine

3.2.1 Liste des oeuvres

Enrichissez la requête pour trouver la liste des œuvres écrites par Jean de La Fontaine.

```
PREFIX dcterms: <http://purl.org/dc/terms/>
SELECT ?oeuvre
WHERE {
  # oeuvres créées par jean de La Fontaine
  ?oeuvre dcterms:creator
  <http://data.bnf.fr/ark:/12148/cb11910267w#about> .
}
```

3.2.2 Et leur titre ?

Comment ramener le titre de ces œuvres ?

```
PREFIX dcterms: <http://purl.org/dc/terms/>
SELECT ?oeuvre ?titreOeuvre
```

```

WHERE {
  # oeuvres créées par jean de La Fontaine
  ?oeuvre dcterms:creator
  <http://data.bnf.fr/ark:/12148/cb11910267w#about> .
  # avec leur titre
  ?oeuvre dcterms:title ?titreOeuvre .
}

```

3.3 Trouver toutes les éditions de ces œuvres

3.3.1 Liste des éditions

Enrichissez la requête pour trouver les éditions des œuvres écrites par Jean de La Fontaine.

```

PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
SELECT ?oeuvre ?titreOeuvre ?edition
WHERE {
  # oeuvres créées par jean de La Fontaine
  ?oeuvre dcterms:creator
  <http://data.bnf.fr/ark:/12148/cb11910267w#about> .
  # avec leur titre
  ?oeuvre dcterms:title ?titreOeuvre .
  # les éditions de ces oeuvres
  ?edition rdarelationshps:workManifested ?oeuvre .
}

```

3.3.2 Et leur titre ?

Comment ramener le titre de ces éditions ? commentez la différence entre titre de l'œuvre et titre de l'édition.

```

PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
SELECT ?oeuvre ?titreOeuvre ?edition ?titreEdition
WHERE {
  # oeuvres créées par jean de La Fontaine
  ?oeuvre dcterms:creator
  <http://data.bnf.fr/ark:/12148/cb11910267w#about> .
  # avec leur titre
  ?oeuvre dcterms:title ?titreOeuvre .
  # les éditions de ces oeuvres
  ?edition rdarelationshps:workManifested ?oeuvre .
  # les titres des éditions
  ?edition dcterms:title ?titreEdition
}

```

3.3.3 Et leur ISBN ?

Comment ramener l'ISBN de ces éditions ? attention, toutes les éditions de ces œuvres n'ont pas un ISBN !

```
PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
PREFIX bnf-onto: <http://data.bnf.fr/ontology/bnf-onto/>
SELECT ?oeuvre ?titreOeuvre ?edition ?titreEdition ?isbn
WHERE {
  # oeuvres créées par jean de La Fontaine
  ?oeuvre dcterms:creator
  <http://data.bnf.fr/ark:/12148/cb11910267w#about> .
  # avec leur titre
  ?oeuvre dcterms:title ?titreOeuvre .
  # les éditions de ces oeuvres
  ?edition rdarelationshps:workManifested ?oeuvre .
  # les titres des éditions
  ?edition dcterms:title ?titreEdition .
  # Les ISBN des éditions - pas toujours présent !
  OPTIONAL { ?edition bnf-onto:isbn ?isbn . }
}
```

3.4 Trouver les versions numériques de ces éditions

3.4.1 La version numérique

Enrichissez la requête pour conservez les versions numériques de ces éditions. Attention, toutes les éditions n'ont pas une version numérique !

Commentez le rapport entre l'ISBN et la version numérique.

```
PREFIX dcterms: <http://purl.org/dc/terms/>
PREFIX rdarelationshps: <http://rdvocab.info/RDARelationshipsWEMI/>
PREFIX bnf-onto: <http://data.bnf.fr/ontology/bnf-onto/>
SELECT ?oeuvre ?titreOeuvre ?edition ?titreEdition ?isbn
?reproductionElectronique
WHERE {
  # oeuvres créées par jean de La Fontaine
  ?oeuvre dcterms:creator
  <http://data.bnf.fr/ark:/12148/cb11910267w#about> .
  # avec leur titre
  ?oeuvre dcterms:title ?titreOeuvre .
  # les éditions de ces oeuvres
  ?edition rdarelationshps:workManifested ?oeuvre .
}
```

```
# les titres des éditions
?edition dcterms:title ?titreEdition .
# Les ISBN des éditions - pas toujours présent !
OPTIONAL { ?edition bnf-onto:isbn ?isbn . }
# les versions électroniques
OPTIONAL { ?edition rdarelationships:electronicReproduction
?reproductionElectronique . }
}
```